
1 行列計算
数値計算の中心課題の一つである，行列に関する演算について見ていく．多次
元，大規模な行列に対する効率のよい計算法が多数開発されており，多くの既存の
ライブラリが用意されている．本章ではそれらの中心をなす，逆行列と固有値に
関して具体的な計算方法を示す．現実的な問題には既存のライブラリを使うのが
上策であるが，それでも基礎となる原理の理解や，ちょっとした計算，ライブラリ
の結果の検証に使えるルーチンを示す．
逆行列は連立一次方程式を解くことと等価である．ルーチン的なやり方にガウス
の消去法がある．これは上三角行列になれば代入を適宜おこなうことで解が容易に
求まることを利用する．さらに，初期値から始めて次々に解に近づけていく反復法が
ある．この代表例であるJacobi(ヤコビ)法と，収束性を高めたGauss-Seidel(ガ
ウス-ザイデル)法を紹介する．
上記の手法をより高速にした修正コレスキー分解と共役傾斜 (共役勾配)法があ
るが，少し複雑になるので割愛する．必要ならばNumRecipeを読め．

2 ガウス消去法による連立一次方程式の解
逆行列は連立一次方程式を解くことと等価である．すなわち，Aを行列，xを未
知数ベクトル，bを数値ベクトルとすると，

Ax = b (1)

A−1Ax = A−1b

x = A−1b

である．未知数の少ない連立一次方程式では，適当に組み合わせて未知数を消し
ていけばいいが，未知数が多くなってしまうと破綻する．未知数の多い多元連立
一次方程式で，ルーチン的に解を求めていく方法がガウス消去法で，前進消去と
後退代入という 2つの操作からなる．

後退代入 (Backward substitution)による解の求め方を先ず見よう．たとえば，

x + y − 2z = −4

−3y + 3z = 9

−z = −2 (2)

では，下から順番に z → y → xと適宜代入することによって，簡単に解を求め
ることが出来る．係数で作る行列でこのような形をした上三角行列にする操作を
前進消去あるいはガウスの消去法 (Gaussian elimination)という．下三角行列
L(lower triangular matrix)と上三角行列 U(upper triangular matrix)の積に分解
する操作

A = L.U (3)

をLU分解 (LU decomposition)という．例えば先に示した上三角行列を係数と
する連立方程式は，

x + y − 2z = −4

x − 2y + z = 5

2x − 2y − z = 2 (4)

を変形することで得られる．この係数行列の U 行列が先程の (2)式の係数行列で
ある．また，この L行列は

(5)

となる．

3 MapleによるLU分解
係数行列 (coefficient matrix)と定数項 (b)との関係は以下の通りである．

> A:=Matrix([[1,1,-2],[1,-2,1],[2,-2,-1]]):

X:=Vector([x,y,z]):

#X:=Vector([1,-1,2]);

b:=Vector([-4,5,2]):

A.X=b;


x + y − 2 z

x − 2 y + z

2 x − 2 y − z

 =


−4

5

2



単に逆行列を求める際は

> with(LinearAlgebra):

MatrixInverse(A);


4/3 5/3 −1

1 1 −1

2/3 4/3 −1



である．Mapleでは行列を三角行列に分解するために，LUDecompositionコマン
ドが用意されている．係数と定数項から作られる行列を拡大係数行列 (augmented

matrix) といい<A|b>のようにして作られる．

> P,L,U:=LUDecomposition(<A|b>):

P , L, U :=


1 0 0

0 1 0

0 0 1

 ,


1 0 0

1 1 0

2 4/3 1

 ,


1 1 −2 −4

0 −3 3 9

0 0 −1 −2



後退代入までおこなってもとめた連立方程式の解は以下の通り．

> LUDecomposition(<A|b>,output=’R’);


1 0 0 1

0 1 0 −1

0 0 1 2

4 MapleによるLU分解のプログラム
LU分解すれば線形方程式の解が容易に求まることは理解できると思う．どのよ
うにして LU分解する行列 (消去行列と称す)の係数が求まるかを以下に示す．ま
ず一般の行列を仮定する．

> with(LinearAlgebra):
A:=Matrix(3,3,AA);

A :=


AA (1, 1) AA (1, 2) AA (1, 3)

AA (2, 1) AA (2, 2) AA (2, 3)

AA (3, 1) AA (3, 2) AA (3, 3)



次のような下三角行列 Lになると仮定する．

> L:=Matrix([[1,0,0],[a,1,0],[b,c,1]]);

L :=


1 0 0

a 1 0

b c 1



これに以下のような行列 T2, T1を

> T1:=Matrix([[1,0,0],[-a,1,0],[-b,0,1]]);

T1 :=


1 0 0

−a 1 0

−b 0 1



> T2:=Matrix([[1,0,0],[0,1,0],[0,-c,1]]);

T2 :=


1 0 0

0 1 0

0 −c 1



右からかけると

> L.T2.T1;

1 0 0

0 1 0

0 0 1



単位行列となる．これをAに掛けてもAから変化しないので，

> L.T2.T1.A;


AA (1, 1) AA (1, 2) AA (1, 3)

AA (2, 1) AA (2, 2) AA (2, 3)

AA (3, 1) AA (3, 2) AA (3, 3)



である．これから，T2.T1.AがU行列となることが分かる．ちょっと難しそうに思
えるが，ひとつずつ解いていくことが可能．具体的にAをLU分解する行列 T1,T2
は次のようにして求められる．

> A0:=Matrix([[1,1,-2],[1,-2,1],[2,-2,-1]]):
b0:=Vector([-4,5,2]):
A:=Matrix(A0):
B:=Vector(b0):
n:=3:
L:=Matrix(array(1..n,1..n,identity)):
for i from 1 to n do
T[i]:=Matrix(array(1..n,1..n,identity)):
for j from i+1 to n do
am:=A[j,i]/A[i,i];
T[i][j,i]:=-am;
L[j,i]:=am;
for k from 1 to n do

A[j,k]:=A[j,k]-am*A[i,k];
end do;
B[j]:=B[j]-B[i]*am;

end do;
end do:

> T[1];


1 0 0

−1 1 0

−2 0 1

> T[2];


1 0 0

0 1 0

0 −4/3 1



> U:=T[2].T[1].A0;

U :=


1 1 −2

0 −3 3

0 0 −1



> L.U;


1 1 −2

1 −2 1

2 −2 −1



> L;A;

(6)

> B;


−4

9

−2

5 ピボット操作
ガウス消去法で困ることに，もしある行を対角要素で割ろうとするときに，その
対角要素が 0になる場合がある．しかし，この場合にも，方程式の順序 (行列の行
と右辺の値)を入れ替えれば解決する．ちなみにこの割るほうの要素をピボット要
素あるいはピボット(pivot,バスケの軸足を動かさずにくるくる回るやつ)と呼ぶ．
この操作は変数の並びは変わらず単に方程式の順番を変更しただけである．
さらに対角要素の数値が厳密に 0でなくとも極端に 0に近づいた場合でも，そ
の数で割った数値が大きくなり他の数との差を取ると以前に示した情報落ちの可
能性が出てくる．この現象を防ぐためには，絶対値最大のピボットを選んで行の
入れ替えを毎回おこなうといい結果が得られることが知られている．

Mapleの LUDecompositionコマンドをこのような行列に適用すると，置換行
列(permutation matrix)Pは単位行列ではなくなる．P.A=L.Uであることに注意．

6 反復法による連立方程式の解
以下のような連立方程式を

5x + y + z + u = −6

x + 3y + z + u = 2

x − 2y − 9z + u = −7

x + 3y − 2z + 5u = 3

形式的に解くと

x = (−6 − ( y + z + u))/5 (7)

y = ( 2 − ( x + z + u))/3

z = (−7 − ( x − 2y + u))/(−9)

u = ( 3 − ( x + 3y − 2z ))/5

となる．適当に初期値 (x0,y0,z0,u0)をとり，下側の方程式に代入すると，得られ
た出力 (x1,y1,z1,u1)はより正解に近い値となる．これを繰り返すことによって正
解が得られる．これをヤコビ (Jacobi)法と呼び，係数行列の対角要素が非対角要
素にくらべて大きいときに適用できる．多くの現実の問題ではこの状況が成り立っ
ている．

Gauss-Seidel法は Jacobi法の高速版である．n番目の解の組が得られた後に一
度に次の解の組に入れ替えるのではなく，得られた解を順次改良した解として使っ

ていく．これにより，収束が早まる．以下にはヤコビ法のコードを示した．x1[i]

の配列を変数に換えるだけで，Gauss-Seidel法となる．

リスト 1: Jacobi法

n:=4;
x0:=[0,0,0,0]:
x1:=[0,0,0,0]:
for iter from 1 to 20 do
for i from 1 to n do
x1[i]:=b[i];
for j from 1 to n do

x1[i]:=x1[i]-AA[i,j]*x0[j];
end do:
x1[i]:=x1[i]+AA[i,i]*x0[i];
x1[i]:=x1[i]/AA[i,i];

end do:
x0:=evalf(x1); #x0を x1に入れ替える
print(iter,x0);
end do:

n := 4
1, [-1.200000000, 0.6666666667, 0.7777777778, 0.6000000000]
2, [-1.608888889, 0.6074074073, 0.5629629630, 0.7511111112]
3, [-1.584296296, 0.7649382717, 0.5474897119, 0.7825185186]
4, [-1.618989300, 0.7514293553, 0.5187050756, 0.6768921810]
5, [-1.589405322, 0.8077973477, 0.5061160189, 0.6804222770]
6, [-1.598867129, 0.8009556753, 0.4972691400, 0.6356490634]
7, [-1.586774776, 0.8219829753, 0.4927633981, 0.6381076766]
8, [-1.590570810, 0.8186345670, 0.4897074389, 0.6212705292]
9, [-1.585922507, 0.8265309473, 0.4881589539, 0.6228163974]

10, [-1.587501260, 0.8249823853, 0.4870924439, 0.6165295146]
11, [-1.585720869, 0.8279597673, 0.4865626093, 0.6173477984]
12, [-1.586374035, 0.8272701537, 0.4861897104, 0.6149933572]
13, [-1.585690644, 0.8283969890, 0.4860087794, 0.6153885990]
14, [-1.585958873, 0.8280977553, 0.4858782197, 0.6145034472]
15, [-1.585695884, 0.8285257353, 0.4858165626, 0.6146844092]
16, [-1.585805341, 0.8283983040, 0.4857707838, 0.6143503606]
17, [-1.585703890, 0.8285613990, 0.4857498236, 0.6144303994]
18, [-1.585748324, 0.8285078890, 0.4857337457, 0.6143038680]
19, [-1.585709101, 0.8285702367, 0.4857266407, 0.6143384296]
20, [-1.585727061, 0.8285480103, 0.4857209840, 0.6142903344]

課題
1. 後退代入法で解を求めよ．

x + 4y − 3z = 1

−6y + 4z = 1

−5/3z = 1/3 (8)

2. 次の行列Aを LU分解せよ． 
1 4 3

1 −2 1

2 −2 −1

 (9)

3. Jacobi法のプログラムを参照してGauss-Seidel法のプログラムを作れ．Jacobi

法と収束性を比べよ．

4. 次の連立方程式の解を求めよ．ただし，pivot操作が必要となる．

>

with(LinearAlgebra):

A:=Matrix([[3,2,2,1],[3,2,3,1],[1,-2,-3,1],[5,3,-2,5]]):

X:=Vector([w,x,y,z]):

b:=Vector([-6,2,-9,2]):

A.X=b;



3 w + 2 x + 2 y + z

3 w + 2 x + 3 y + z

w − 2 x − 3 y + z

5 w + 3 x − 2 y + 5 z


=



−6

2

−9

2


5. pivot操作を含めたLU分解のプログラムを作成せよ．上の問題を解き，その

L,U 行列および L−1.bベクトルを求めよ．

